Guide to recreate RNTV-Index.xlsm

Steps

- 1A Enter Raw Data
- 1B Transform and Check Data
- **2A** Random Book
- 2B Random Chapter and Verse
- **2C** Add Button

Page | 1

1A Gather and enter data into the spreadsheet

Use a bible to obtain the list of books in the New Testament, number of chapters in each book, and number of verses in each chapter. The data here are from the RSV-2CE bible.

Save input sheet as comma delimited text file (RNTV-RawData.csv) (File >> Export >> Change File Type)

(Sheet 1A Raw Data)

1B Prepare data and insert helper columns

The helper columns will identify the randomly chosen book, chapter and verse.

Use autofill to number the books from 1...27 in cells B5...B31.

Insert a new column before column D; choose Format same as right from the InsertOptions.

Select column F and convert Text to Columns (Data tab).

In the Wizard, choose Delimited conversion with Comma and Space as delimiters, and check Treat consecutive delimiters as one. Specify F4 as destination.

Select columns F to AI to Autofit Column Width (Home tab >> Format), and remove left { and right } curly braces (Home tab >> Find & Select >> Replace { with blank).

Select G5...AH31, and insert 0 in blank cells (Ctrl-F >> Replace blank with 0).

To check integrity of transformed data, use COUNTIF to verify that the number of verses is equal to number of chapters for each book.

In cell AJ5 type =IF(COUNTIF(\$G5:\$AH5,"> 0")=\$E5,"T",FALSE).
"T" is used instead of TRUE, so that a FALSE result will stand out.

Use Autofill to copy this copy down to AJ31.

If the results are all "T", the integrity of the prepared data is preserved by the transformation, and the contents of the AJ column can be cleared. Otherwise, check and correct data at rows indicated FALSE.

To trigger FALSE results, change a number in the Chapters column, replace a 0 with a number in one of the Verses columns and vice versa. Be sure to undo these changes.

To improve readability, center content for columns B, D and F, and add column headings accordingly.

(Sheet 1B Prepared Data)

Page | 2

2A Choose a book at random

Use the RANDBETWEEN function to pick one of the 27 books in the NT. In cell B3, type =RANDBETWEEN(1,27).

Use the index function to identify the randomly selected book. In C3, type =INDEX(\$C\$5:\$C\$31, \$B\$3).

Use of absolute reference (F4 shortcut) in formulas makes it easier to copy or move this cell to another location in the sheet, without compromising its content.

Use <u>conditional formatting</u> to highlight book number to make it easier to see which book is selected. (Home tab >> Conditional Formatting >> Highlight Cell Rules >> Equal To)
Cell value equals to \$B\$3, apply rule to \$B\$5:\$B\$31 (Conditional Formatting Rules Manager)

(Sheet 2A Random Book)

2B Select random chapter and random verse

Select a random chapter, from the book chosen at random from the previous step.

In D5, type =IF(\$B\$3=\$B5,RANDBETWEEN(1,\$E5),""). Note the use of mixed reference in the formula.

Use Autofill to copy this formula down to D27.

Pick a random verse, from the chapter selected at random from the previous step.

In F5, type =IF(ISNUMBER(\$D5), RANDBETWEEN(1, INDEX(\$G5:\$AH5,1,\$D5)),"").

Use Autofill to copy this formula down to F27.

In D3, type =MAX(\$D\$5:\$D\$31). In F3, type =MAX(\$F\$5:\$F\$31).

To check that the picked verse exists in the selected chapter and chosen book, in AJ5 type =IFS(NOT(ISNUMBER(\$D5)), "", \$D5 > \$E5, FALSE, \$F5 > INDEX(\$G5:\$AH5, 1, \$D5), FALSE, TRUE, "T").

IFS conditions are evaluated one after another; this evaluation stops when a condition evaluates to true. The first condition is true when a book is not chosen, in which case nothing is displayed.

The second condition is true when a selected chapter is not in the chosen book, in which case FALSE is displayed.

The third condition is true when a picked verse is not in the selected chapter, in which case FALSE is displayed.

The fourth condition is evaluated when none of the previous conditions evaluated to true, in which case a "T" is displayed.

In a correctly working sheet, column AJ displays a single "T" at the row highlighted by conditional formatting.

With cell D5 selected, use conditional formatting to highlight selected chapter and verse (Home tab >> Conditional Formatting >> New Rule). Select as Rule Type, Format only cells that contain No Blanks. Apply rule to cell range \$D\$5:\$F\$31.

Page | 3

Test the sheet formulas: insert 7 into D13, and insert 1 and 15 into D24 and F24 respectively. Both tests should trigger a FALSE in AJ on the corresponding row. Be sure to restore the formulas in these cells after testing.

Hide columns E, and G to AJ (select column >> right-click >> Hide).

(Sheet 2B Random Chapter and Verse)

2C Add user control (macro and button) to generate new random verse

Name cells C3, D3 and F3 respectively as Book, Chapter, and Verse.

In cell AL3 type =CONCAT(Book, " ",Chapter, ":", Verse).

Format result cell AL3 according to design taste. Use <u>separator</u> to widen column AL. Bold (Ctrl-B) content of AL3, change font colour.

Ensure dependent cells are updated automatically (Formulas >> Calculation Options >> Automatic).

Configure macro settings for workbook:

(File >> Options >> TrustCentre(Settings) >> MacroSettings >> Disable all macros with notifications).

Add Developer tab (Right click ribbon >> Customize the Ribbon).

Record macro named NewVerse. Select cell B3, press F2 key, and hit Enter. Stop macro recording. View Code for macro (Module 1). Change B4 to E3.

Draw a button (Developer tab >> Insert >> Button) around cell AL6. Assign it the NewVerse macro. Relabel it 'New Verse' (Right-click on button >> Edit Text). Test the button.

(Sheet 2C Add Button)

Save file as a macro-enabled workbook (.xlsm).

Format the button control so it moves but does not resize with cells.

Hide other columns or rows according to design taste.

Link to an online Bible source to obtain the text for the random verse. In cell AL4, type =HYPERLINK (CONCAT("https://www.biblegateway.com/passage/?search=", \$AL\$3,"&version=RSVCE"), "Read Verse here")

Test that the link retrieves the correct verse.

Optional: Read the verse and lookup its surrounding text for context.

Page | 4